Problem: Let $ABCD$ be a cyclic quadrilateral with $AB=7$ and $CD=8$.
Points $P$ and $Q$ are selected on line segment $AB$ so that $AP=BQ=3$.
Points $R$ and $S$ are selected on line segment $CD$ so that $CR=DS=2$.
Prove that $PQRS$ is a cyclic quadrilateral.

Solution: Here are three possible approaches.

\paragraph{The one-liner.}
The four points $P$, $Q$, $R$, $S$ have equal power $-12$ with respect to $(ABCD)$.
So in fact they're on a circle concentric with $(ABCD)$.

\paragraph{The external power solution.}
We distinguish between two cases.

\subparagraph{Case where $AB$ and $CD$ are not parallel.}
We let lines $AB$ and $CD$ meet at $T$.
Without loss of generality,
$A$ lies between $B$ and $T$ and $D$ lies between $C$ and $T$.
Let $x=TA$ and $y=TD$, as shown below.
\begin{center}
\begin{asy}
size(8cm);
pair A = dir(170);
pair B = dir(80);
pair C = dir(323.912853);
pair D = dir(216.087147);
pair T = extension(A, B, C, D);
filldraw(unitcircle, opacity(0.1)+lightcyan, blue);
draw(B--T--C--B, blue);
draw(A--D, blue);

pair P = 4*A/7+3*B/7;
pair Q = 3*A/7+4*B/7;
pair R = 6*C/8+2*D/8;
pair S = 2*C/8+6*D/8;

filldraw(circumcircle(P, Q, R), opacity(0.1)+yellow, dashed+red);
label("$x$", midpoint(T--A), dir(A+B), deepgreen);
label("$3$", midpoint(A--P), dir(A+B), deepgreen);
label("$1$", midpoint(P--Q), -dir(A+B), deepgreen);
label("$3$", midpoint(Q--B), dir(A+B), deepgreen);
label("$y$", midpoint(T--D), dir(C+D), deepgreen);
label("$2$", midpoint(D--S), dir(C+D), deepgreen);
label("$4$", midpoint(S--R), -dir(C+D), deepgreen);
label("$2$", midpoint(R--C), dir(C+D), deepgreen);

dot("$A$", A, dir(120));
dot("$B$", B, dir(B));
dot("$C$", C, dir(300));
dot("$D$", D, dir(240));
dot("$T$", T, dir(T));
dot("$P$", P, dir(160));
dot("$Q$", Q, dir(80));
dot("$R$", R, dir(270));
dot("$S$", S, dir(270));

/* --------------------------------+
| TSQX: by CJ Quines and Evan Chen |
| https://github.com/vEnhance/tsqx |
+----------------------------------+
A 120 = dir 170
B = dir 80
C 300 = dir 323.912853
D 240 = dir 216.087147
T = extension A B C D
unitcircle / 0.1 lightcyan / blue
B--T--C--B / blue
A--D / blue
P 160 = 4*A/7+3*B/7
Q 80 = 3*A/7+4*B/7
R 270 = 6*C/8+2*D/8
S 270 = 2*C/8+6*D/8
circumcircle P Q R / 0.1 yellow / dashed red
!label("$x$", midpoint(T--A), dir(A+B), deepgreen);
!label("$3$", midpoint(A--P), dir(A+B), deepgreen);
!label("$1$", midpoint(P--Q), -dir(A+B), deepgreen);
!label("$3$", midpoint(Q--B), dir(A+B), deepgreen);
!label("$y$", midpoint(T--D), dir(C+D), deepgreen);
!label("$2$", midpoint(D--S), dir(C+D), deepgreen);
!label("$4$", midpoint(S--R), -dir(C+D), deepgreen);
!label("$2$", midpoint(R--C), dir(C+D), deepgreen);
*/
\end{asy}
\end{center}
By power of a point,
\begin{align*}
  \text{$ABCD$ cyclic} \iff x(x+7) &= y(y+8) \\
  \text{$PQRS$ cyclic} \iff (x+3)(x+4) &= (y+2)(y+6).
\end{align*}
However, the latter equation is just the former with $12$ added to both sides.
(That is, $(x+3)(x+4) = x(x+7)+12$ while $(y+2)(y+6)=y(y+8)+12$.)
So the conclusion is immediate.

\subparagraph{Case where $AB$ and $CD$ are parallel.}
In that case $ABCD$ is an isosceles trapezoid.
Then the entire picture is symmetric around the common perpendicular bisector
of the lines $AB$ and $CD$.
Now $PQRS$ is also an isosceles trapezoid, so it's cyclic too.
\begin{center}
\begin{asy}
size(4cm);
pair A = dir(135);
pair B = dir(45);
pair C = dir(323.912853);
pair D = dir(216.087147);
filldraw(unitcircle, opacity(0.1)+lightcyan, blue);
draw(A--B--C--D--cycle, blue);

pair P = 4*A/7+3*B/7;
pair Q = 3*A/7+4*B/7;
pair R = 6*C/8+2*D/8;
pair S = 2*C/8+6*D/8;

filldraw(circumcircle(P, Q, R), opacity(0.1)+yellow, dashed+red);

dot("$A$", A, dir(120));
dot("$B$", B, dir(B));
dot("$C$", C, dir(300));
dot("$D$", D, dir(240));
dot("$P$", P, dir(90));
dot("$Q$", Q, dir(80));
dot("$R$", R, dir(270));
dot("$S$", S, dir(270));
\end{asy}
\end{center}

\paragraph{The Pythagorean bash.}
Let $\rho$ and $O$ denote the radius and center of the circle circumscribing $ABCD$.
We will show that in fact, $P$, $Q$, $R$, $S$ lie on a circle centered at $O$,
i.e.\ that the lengths $PO$, $QO$, $RO$, $SO$ are all equal.

\begin{center}
\begin{asy}
size(9cm);
pair A = dir(170);
pair B = dir(80);
pair C = dir(323.912853);
pair D = dir(216.087147);
filldraw(unitcircle, opacity(0.1)+lightcyan, blue);
draw(A--B--C--D--cycle, blue);

pair P = 4*A/7+3*B/7;
pair Q = 3*A/7+4*B/7;
pair R = 6*C/8+2*D/8;
pair S = 2*C/8+6*D/8;
pair O = origin;
pair M = midpoint(P--Q);
pair N = midpoint(R--S);
draw(N--O--M, dashed);

filldraw(circumcircle(P, Q, R), opacity(0.1)+yellow, dashed+red);
label("$3$", midpoint(A--P), dir(A+B), deepgreen);
label("$3$", midpoint(Q--B), dir(A+B), deepgreen);
label("$2$", midpoint(D--S), dir(C+D), deepgreen);
label("$2$", midpoint(R--C), dir(C+D), deepgreen);

label("$0.5$", midpoint(P--M), dir(285), deepgreen);
label("$0.5$", midpoint(Q--M), dir(330), deepgreen);
label("$2$", midpoint(S--N), -dir(C+D), deepgreen);
label("$2$", midpoint(R--N), -dir(C+D), deepgreen);


dot("$A$", A, dir(120));
dot("$B$", B, dir(B));
dot("$C$", C, dir(300));
dot("$D$", D, dir(240));
dot("$P$", P, dir(160));
dot("$Q$", Q, dir(80));
dot("$R$", R, dir(270));
dot("$S$", S, dir(270));
dot("$O$", O, dir(0));
dot("$M$", M, dir(M));
dot("$N$", N, dir(N));
\end{asy}
\end{center}

Our proof will use repeated applications of the Pythagorean theorem.
Let $M$ denote the midpoint of $\overline{PQ}$,
which is also the midpoint of $\overline{AB}$,
as $AM = MB = 3.5$ and $PM = QM = 0.5$.
Assuming $M$ is distinct from $O$,
it follows $\overline{OM}$ is the perpendicular bisector of $\overline{AB}$,
and so the Pythagorean theorem we get
\begin{align*}
  PO^2 &= PM^2 + MO^2 = PM^2 + (AO^2 - AM^2) = 0.5^2 + \rho^2 - 3.5^2 \\
  QO^2 &= QM^2 + MO^2 = QM^2 + (BO^2 - BM^2) = 0.5^2 + \rho^2 - 3.5^2.
\end{align*}
And in the case $O = M$, the same equations are valid too.

Similarly let $N$ denote the midpoint of $\overline{RS}$,
which is also the midpoint of $\overline{CD}$,
as $CN = DN = 4$ and $RN = SN = 2$.
Repeating the same calculation gives
\begin{align*}
  RO^2 &= RN^2 + NO^2 = RN^2 + (CO^2 - CN^2) = 2^2 + \rho^2 - 4^2 \\
  SO^2 &= SN^2 + NO^2 = SN^2 + (DO^2 - DN^2) = 2^2 + \rho^2 - 4^2.
\end{align*}
From this it follows that
\[ PO^2 = QO^2 = RO^2 = SO^2 = \rho^2 - 12. \]
Hence the problem is solved.